


◀ (<http://vbhack.ru>) Урок VBScript №9: Строковые функции. Работа с ТЕКСТОМ

📅 15 сентября, 2015 📁 Уроки VBScript (<http://vbhack.ru/category/uroki-vbscript>) 🔍

[vbs](http://vbhack.ru/tag/vbs) (<http://vbhack.ru/tag/vbs>) , [VBScript](http://vbhack.ru/tag/vbscript) (<http://vbhack.ru/tag/vbscript>) ,

[Работа с текстом](http://vbhack.ru/tag/rabota-s-tekstom) (<http://vbhack.ru/tag/rabota-s-tekstom>) , [Строковые](http://vbhack.ru/tag/strokovye) (<http://vbhack.ru/tag/strokovye>) ,

[урок](http://vbhack.ru/tag/urok) (<http://vbhack.ru/tag/urok>) , [Функции](http://vbhack.ru/tag/funkcii) (<http://vbhack.ru/tag/funkcii>)



Сегодня я хотел бы познакомить вас с интересной статьёй про **строковые функции VBScript**. Этот материал будет очень интересен и полезен, так как мы можем любую используемую информацию превратить в текст и произвести с ним нужные нам действия при помощи данных функций. Благодаря им мы можем создавать свой **парсер (грабер) html кода веб страницы**, текстовый редактор и многое другое. Ну что же, приступим.

Функции Instr и InstrRev

Функция **Instr** позволяет получить номер символа (позицию) с которого начинается искомая фраза или другая последовательность символов.

Синтаксис функции: **Instr([Начальный Символ,] Строка1, Строка2 [, Сравнение])**

Начиная с этого урока, я буду помещать не обязательные параметры в квадратные скобки. А теперь рассмотрим их получше:

- **Начальный Символ** — номер символа, с которого надо начинать искать. При этом отчёт всё равно будет идти от первого символа.
- **Строка1** — строка в которой ищем
- **Строка2** — строка которую ищем
- **Сравнение** — учитывать или не учитывать регистр. Да — 0 = **vbBinaryCompare**, Нет — 1 = **vbTextCompare**. По умолчанию 0.

Функция **InstrRev** работает аналогично, но с конца строки. При этом возвращаемый результат отсчитывается от начала строки.

Синтаксис: **Instr(Строка1, Строка2[, Начальный Символ] [, Сравнение])**

Теперь посмотрим примеры обеих функций

</>



```
'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_1.vbs
'*****

Dim a, b
a = "Строка в которой что-то ищут"
b = "Что-то"

MsgBox InStr(19,a,b) & vbCrLf & InStr(3,a,b) & vbCrLf & InStr(3,a,b,1) & vbCrLf & InStr(19,a,b,1)

MsgBox InStrRev(a,b,23) & vbCrLf & InStrRev(a,b,23,1) & vbCrLf & InStrRev(a,b,-1,1)
```

В данном примере четыре варианта использования функции **InStr**. Стоит отметить, что если вы указываете **сравнение**, то будет обязательным использование аргумента **Начальный Символ**.

Функция Mid

Данная функция vbs позволяет вырезать из строки определённое количество символов.

Синтаксис: **Mid(строка, начало[, длина])**

- **Строка** — Понятно. Строка, с которой работаем.
- **Начало** — Номер символа, с которого начинаем
- **Длина** — Сколько символов вырезаем. Не обязательно.

</>



```
'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_2.vbs
'*****

Dim a, b, c
a = "Строка в которой что-то ищут"
b = "Что-то"
c = InStr(3,a,b,1)

Msgbox Mid(a,c) & vbCrLf & Mid(a,c,3)
```

В данном примере я так же использовал предыдущую функцию. При помощи данных функций легко создать парсер html кода веб страниц. Я активно их использовал при создании **парсера почт групп Мой Мир Mail.ru**.

Функции LCase и UCase

Данные функции позволяют преобразовать все символы строки к нижнему (**LCase**) и верхнему (**UCase**) регистру. Данные функции могут быть полезны в разных ситуациях. Например, когда вы создаёте диалоговое окно `InputBox` куда надо что то вписать, а потом найти совпадение в условных операторах.

```
</>
'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_3.vbs
'*****

Dim a, b
a = "СтРоКа 1"
b = "СтРоКа 2"

Msgbox LCase(a) & vbCrLf & UCase(b)
```

Функции Left и Right

Данные **функции VBScript** возвращают нам определённое количество символов с начала (**Left**) и конца (**Right**) строки.

```
</>
'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_4.vbs
'*****

Dim a
a = "СтРоКа 1"

Msgbox Left(a,3) & vbCrLf & Right(a,3)
```

Функции LTrim, RTrim и Trim

Данные функции убирают пробелы вначале (**LTrim**), в конце (**RTrim**) и с обеих сторон (**Trim**) строки.

```
</>
```

```

'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_5.vbs
'*****

Dim a, b
a = "      СтРоКа      "
b = "!"
Msgbox LTrim(a) & b & vbCrLf & RTrim(a) & b & vbCrLf & Trim(a) & b

```

Функция Replace

Это очень интересная функция. Она позволяет производить замену указанных символов на другие.

Синтаксис: **Replace(Строка, Что ищем, Чем заменяем [, Начало[, Кол. Замен[, Сравнение]])**

- **Строка** — Строка, в которой ищем
- **Что ищем** — Комбинация символов, которые надо заменить.
- **Чем заменить** — Понятно.
- **Начало** — с какого по счёту символа начинать поиск.
- **Кол. Замен** — количество замен, которые нужно сделать. По умолчанию заменит все совпадения (-1).
- **Сравнение** — учитывать или не учитывать регистр. Да — 0 = **vbBinaryCompare**, Нет — 1 = **vbTextCompare**. По умолчанию 0.

</>



```

'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_6.vbs
'*****

Dim a
a = "VBHack.Ru лучший сайт!"

Msgbox Replace(a,"лучший","самый лучший")

```

Функции Space и String

Space — формирует строку из указанного количества пробелов.

String — формирует строку из указанного символа, повторяющегося указанное количество раз.

</>



```

'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_7.vbs
'*****

Dim a, b
a = Space(10)
b = String(3,"V")

Msgbox a & b & vbCrLf & b

```

Функция StrComp

Эта функция сравнивает строки и возвращает три результата.

Синтаксис: **StrComp(строка 1, строка 2[, сравнение])**

Возвращаемые результаты:

- **-1** — Если строка 1 меньше строки 2
- **0** — Если строки равны(по символам и их количеству)
- **1** — Если строка 1 больше строки 2

</>



```

'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_8.vbs
'*****

Dim a, b, c
a = "строка"
b = "строка"
c = StrComp(a,b)

If c = 0 then
    MsgBox "Строки совпадают"
elseif c = 1 then
    MsgBox "Строка а больше строки b"
elseif c = -1 then
    MsgBox "Строка а меньше строки b"
end if

```

Функция StrReverse

Забавная функция vbs которая меняет порядок символов на противоположный (задом на перёд)

</>



```
'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_9.vbs
'*****

Dim a
a = "унитаз"

MsgBox StrReverse(a)
```

Функция Len

Данная функция возвращает количество символов в строке.

```
</>
'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_10.vbs
'*****

Dim a
a = "унитаз"

MsgBox Len(a)
```

Функции Asc и Chr

Asc — Возвращает ASCII-код первого символа в строке.

Chr — Возвращает символ по указанному ASCII-коду.

```
</>
'*****
'Урок VBScript №9:
'Строковые функции. Работа с текстом
'file_11.vbs
'*****

Dim a, b, c

a = "унитаз"
b = Asc(a)
c = Chr(b)

MsgBox "ANSI " & c & ": " & b
```

На этом у меня всё! До новых встреч!



 Reply

 Preview

 Help

Markdown enabled

Name

E-Mail

URL

Код безопасности *



Введите символы отображаемые выше:

Post comment

 [Back to top](#)

© VBHack

Theme by nehalist.io (<http://nehalist.io>)