


◀ (<http://vbhack.ru>) Урок VBScript №7: Функции для работы с датой и временем

📅 11 августа, 2015 📁 Уроки VBScript (<http://vbhack.ru/category/uroki-vbscript>) 🔖

[Date](http://vbhack.ru/tag/date) (<http://vbhack.ru/tag/date>) , [DateAdd](http://vbhack.ru/tag/dateadd) (<http://vbhack.ru/tag/dateadd>) , [DateDiff](http://vbhack.ru/tag/datediff) (<http://vbhack.ru/tag/datediff>) ,
[DatePart](http://vbhack.ru/tag/datepart) (<http://vbhack.ru/tag/datepart>) , [DateValue](http://vbhack.ru/tag/datevalue) (<http://vbhack.ru/tag/datevalue>) ,
[FormatDateTime](http://vbhack.ru/tag/formatdatetime) (<http://vbhack.ru/tag/formatdatetime>) , [Hour](http://vbhack.ru/tag/hour) (<http://vbhack.ru/tag/hour>) ,
[Minute](http://vbhack.ru/tag/minute) (<http://vbhack.ru/tag/minute>) , [Month](http://vbhack.ru/tag/month) (<http://vbhack.ru/tag/month>) , [MonthName](http://vbhack.ru/tag/monthname) (<http://vbhack.ru/tag/monthname>) ,
[Time](http://vbhack.ru/tag/time) (<http://vbhack.ru/tag/time>) , [Timer](http://vbhack.ru/tag/timer) (<http://vbhack.ru/tag/timer>) , [TimeSerial](http://vbhack.ru/tag/timeserial) (<http://vbhack.ru/tag/timeserial>) ,
[TimeValue](http://vbhack.ru/tag/timevalue) (<http://vbhack.ru/tag/timevalue>) , [vbs](http://vbhack.ru/tag/vbs) (<http://vbhack.ru/tag/vbs>) , [VBScript](http://vbhack.ru/tag/vbscript) (<http://vbhack.ru/tag/vbscript>) ,
[Weekday](http://vbhack.ru/tag/weekday) (<http://vbhack.ru/tag/weekday>) , [WeekdayName](http://vbhack.ru/tag/weekdayname) (<http://vbhack.ru/tag/weekdayname>) ,
[Year](http://vbhack.ru/tag/year) (<http://vbhack.ru/tag/year>) , [урок](http://vbhack.ru/tag/urok) (<http://vbhack.ru/tag/urok>) , [Функции](http://vbhack.ru/tag/funkcii) (<http://vbhack.ru/tag/funkcii>)



В этом уроке мы пройдем все **функции VBScript для работы с датой и временем**. Данные функции VBS дают нам возможность получить текущую дату и время, временные промежутки, создавать таймеры и многое другое, что не оставит Вас равнодушными.

Для того что бы получить текущую дату вашего компьютера используется **функция «Date»**, а если Вам понадобится по мимо даты вывести ещё и время, то надо использовать **функцию «Now»**. Для возврата только времени — **«Time»**.

Так же любой переменной мы можем задать свою дату. Для этого используется символ решётка (**#**). В данном случае дата прописывается в таком порядке: мм/дд/гггг. Можно так же присвоить время: чч.мм.сс . Давайте посмотрим на пример:

</>



```
'Урок VBScript №7:
'Функция для работы с датой и временем
'file_1.vbs

Dim my_date, my_date1, my_date2, my_date3

my_date = Date
my_date1 = Now
my_date2 = #12/05/2010#
my_date3 = #15.05.10#

MsgBox my_date & vbCrLf & my_date1 & vbCrLf & my_date2 & vbCrLf & my_date3
```

В результате у нас получилась интересная картина. Для переменной **«my_date2»** мы указали свою дату. Её мы записали, начиная с месяца, но когда мы получили значение переменной, то оказалось, что дата была показана учитывая настройки вашего компьютера. У меня в данном случае показало так: день-месяц-год.

С этим мы разобрались, а теперь рассмотрим **функцию vbs — DateDiff**. Она позволяет получить значение интервала двух дат. Для этого нам помогут следующие значения интервала:

- **yyyу** — год
- **q** — квартал
- **m** — месяц
- **y** — номер дня в году
- **d** — день
- **w** — номер дня в неделе
- **ww** — номер недели в году
- **h** — час
- **n** — минута
- **s** — секунда

Синтаксис функции: **DateDiff(«Interval», Date1, Date2, FirstDayOfWeek , FirstWeekOfYear)**.

- **Interval** — Показывает тип интервала, который мы хотим получить. Значения показаны выше.
- **Date1, Date2** — Две даты, между которыми вычисляется разница.
- **FirstDayOfWeek** — Необязательный параметр. Определяет, с какого дня следует считать начало недели.
- **FirstWeekOfYear** — Необязательный параметр. Указывает на ту неделю, которую следует считать первой в году.

Давайте посмотрим пример данной функции VBScript:

```
</>
```



```
'Урок VBScript №7:
'Функция для работы с датой и временем
'file_2.vbs

Dim my_date, my_date1

my_date = #08/03/2015#
my_date1 = #08/06/2015#

'Прошло 72 часа
MsgBox DateDiff ("h", my_date, my_date1)

'Прошло 3 дня
MsgBox DateDiff ("d", "03/08/2015", "06/08/2015")
```

В данном примере всё понятно. Следует обратить внимание на то, что в качестве параметров Date1 и Date2 можно указывать строки. Во втором примере мы так и сделали и, VBS сам распознал, что мы указали дату.

Параметр FirstWeekOfYear может иметь такие значения:

- **vbFirstJan1** — неделя в которой есть 1 января.
- **vbFirstFourDays** — использовать неделю, в которой как минимум четыре дня приходятся на новый год.
- **vbFirstFullWeek** — первая полная неделя в году

Функция DateAdd

DateAdd позволяет добавить к заданной дате определённый промежуток или убрать этот промежуток.

Выглядит всё это дело так: **DateAdd(«Interval», Number, Date)**

- **Interval** — мы проходили с функцией DateDiff.
- **Number** — Число, которое надо добавить или вычесть.
- **Date** — Дата, с которой работаем.

Вот простенький пример:

```
</>
```



```
'Урок VBScript №7:
'Функция для работы с датой и временем
'file_3.vbs

Dim my_date

my_date = #08/03/2015#

' Добавляем 8 месяцев
MsgBox DatAdd("m", 8, my_date)

' Убираем 8 месяцев
MsgBox DateAdd("m", -8, my_date)
```

Функция DatePart

DatePart вернёт нам определённую часть даты: месяц, год, квартал, день и так далее.

Синтаксис функции: **DatePart**(«Interval», Date, FirstDayOfWeek , FirstWeekOfYear)

</>



```
'Урок VBScript №7:
'Функция для работы с датой и временем
'file_4.vbs

Dim my_date

my_date = #08/03/2015#

MsgBox DatePart("m", my_date)
MsgBox DatePart("d", my_date)
MsgBox DatePart("q", my_date)
```

Функции: Hour, Minute, Month, Year, **Second**, TimeValue

- **Hour** — данная функция vbs даёт возможность получить число часов из указанного времени (от 0 до 23).
- **Minute** — позволяет получить число минут из указанного времени (от 0 до 59).
- **Month** — вернёт месяц указанной даты (от 1 до 12).
- **Day** — возвращает номер дня указанного месяца (от 1 до 31).
- **Year** — вернёт год указанной даты.
- **Second** — вернёт количество секунд, указанной даты (от 0 до 59).
- **TimeValue** — возвращает время указанной даты.

Сразу посмотрим пример:

</>



```
'Урок VBScript №7:  
'Функция для работы с датой и временем  
'file_5.vbs
```

```
Dim my_date, my_date2
```

```
my_date = #08/03/2015 14:30#  
my_date2 = "13:49"
```

```
MsgBox "Часы: " & Hour(my_date)  
MsgBox "Минуты: " & Minute(my_date)  
MsgBox "Месяц: " & Month(my_date)  
MsgBox "День: " & Day(my_date)  
MsgBox "Год: " & Year(my_date)  
MsgBox "Время: " & TimeValue(my_date)
```

```
MsgBox "Часы 2: " & Hour(my_date2)  
MsgBox "Минуты 2: " & Minute(my_date2)
```

Функция Timer

Данная функция возвращает количество секунд, прошедших с полуночи, с точностью до 1 миллисекунды. Немного смекалки и данная функция vbs позволяет нам узнать время выполнения кода с поразительной точностью. Давайте посмотрим пример с циклом While ... Wend (<http://vbhack.ru/uroki-vbscript/urok-vbscript-n4-cikly-while-wend-i-do-loop.html>):

</>



```
'Урок VBScript №7:  
'Функция для работы с датой и временем  
'file_6.vbs
```

```
Dim old_time, my_timer, i
```

```
old_time = Timer
```

```
i = 0  
While i < 10000000  
    i = i + 1  
Wend
```

```
my_timer = Timer - old_time  
MsgBox "Цикл был выполнен за " & my_timer & " секунд."
```

Перед началом цикла мы узнали, сколько секунд прошло с полуночи и сохранили это число в переменную «**old_time**». После завершения цикла мы ещё раз узнали сколько прошло секунд с полуночи, но по мимо этого, ещё и вычили предыдущее значение. Таким способом мы получили время, в секундах, затраченное на выполнения **цикла vbs**.

Функция TimeSerial и DateSerial

Функция TimeSerial позволяет преобразовать указанные значения в дату и время. Для этой функции указывают три значения через запятую, в круглых скобках: часы (от 0 до 23), минуты (от 0 до 59), секунды (от 0 до 59). Если какое либо значение превышает допустимый лимит, то это отражается на следующем значении.

Функция DateSerial — делает тоже самое, но только: год, месяц, день.

```
</>
'Урок VBScript №7:
'Функция для работы с датой и временем
'file_7.vbs

Dim My_Time, My_Time_2, My_Time_3, my_h

My_h = 66

My_Time = TimeSerial(10,25,25)
My_Time_2 = TimeSerial(My_h,55,85)

My_Time_3 = DateSerial(2016,12,23)

MsgBox My_Time & vbCrLf & My_Time_2 & vbCrLf & My_Time_3
```

Здесь всё понятно.

Функция Weekday

Weekday позволяет нам получить номер дня недели указанной даты. Данная функция имеет такой вид: **Weekday(date, FirstDayOfWeek)**, где **date** — это используемая дата; **FirstDayOfWeek** — это день недели, который нужно считать началом недели. Последний параметр не обязателен и может иметь такие значения:

- **vbUseSystemDayOfWeek** — использовать настройки по умолчанию системы
- **vbSunday** — воскресенье
- **vbMonday** — понедельник
- **vbTuesday** -вторник

- **vbWednesday** — среда
- **vbThursday** — четверг
- **vbFriday** — пятница
- **vbSaturday** — суббота

Вместо этих выражений можно использовать цифры (от 0 до 7).

Смотрим пример:

```
</>
'Урок VBScript №7:
'Функция для работы с датой и временем
'file_8.vbs

Dim my_date, number_day

number_day = WeekDay(#14/02/1995#, vbSunday)

MsgBox number_day
```

В данном случае я указал начало недели с воскресенья и по этой причине вторник оказался третьем днём недели.

Функция WeekdayName

Замечательная **функция VBScript**. Она возвращает наименование указанного дня недели.

Функция имеет данный синтаксис: **WeekdayName(Number, Abbr, FirstDayOfWeek)**.

- **Number** — Номер дня недели
- **Abbr** — использование или не использование аббревиатуры. Если True, то вернёт аббревиатуру, в противном случае — полное название
- **FirstDayOfWeek** — уже выяснили.

```
</>
'Урок VBScript №7:
'Функция для работы с датой и временем
'file_9.vbs

Dim my_date
my_date = WeekdayName(3, False, vbMonday)
MsgBox my_date
my_date = WeekdayName(3, True, vbMonday)
MsgBox my_date
```

Функция DateValue

Данная простенькая функция возвращает указанное значение имеющее тип Date. Ничем особым она не отличается.

```
</>  
  
'Урок VBScript №7:  
'Функция для работы с датой и временем  
'file_10.vbs  
  
dim my_date  
  
my_date = date  
  
MsgBox DateValue(my_date)  
MsgBox DateValue("14, февраль, 1995")  
MsgBox DateValue("02,14,1995")  
MsgBox DateValue("02/14/1995")
```

Функция MonthName

Возвращает название указанного месяца. Так же можно сделать показ аббревиатуры как в функции **WeekdayName**.

```
</>  
  
'Урок VBScript №7:  
'Функция для работы с датой и временем  
'file_11.vbs  
  
MsgBox MonthName(9,true)  
MsgBox MonthName(9,false)
```

Стоит обратить внимание на то, что месяц май не сокращается, а пишется лишь с маленькой буквы.

Функция FormatDateTime

Данная **функция VBScript** позволяет устанавливать формат даты и времени. Всего форматов 5. Описывать я их не буду, лучше взгляните сами.

```
</>
```



```
'Урок VBScript №7:
'Функция для работы с датой и временем
'file_12.vbs

dim my_date, my_date_2, my_date_3, my_date_4, my_date_5
my_date = FormatDateTime(Now, 0) 'формат "по умолчанию"
my_date_2 = FormatDateTime(Now, 1)
my_date_3 = FormatDateTime(Now, 2)
my_date_4 = FormatDateTime(Now, 3)
my_date_5 = FormatDateTime(Now, 4)

MsgBox my_date & vbCrLf & my_date_2 & vbCrLf & my_date_3 & vbCrLf & my_date_4 & vbCrLf & my_date_5
```

Здесь всё понятно.

Функция IsDate

Данная функция проверяет корректность указанной даты. Если дата указана корректно, то функция вернёт True и False в противном случае.

```
</>
'Урок VBScript №7:
'Функция для работы с датой и временем
'file_13.vbs

Dim a

a = "10/14/2012"

If IsDate(a) then
    MsgBox "Дата корректна!"
else
    MsgBox "Дата не корректна!"
end if
```

На этом наш урок подходит к концу. *До новых встреч на VBHack.Ru!*

2 comments



Kotik24 2 года ago

reply (<http://vbhack.ru/uroki-vbscript/urok-vbscript-n7-funkcii-dlya-raboty-s-datoy-i-v.html?replytocom=21367#respond>)

Нашел ошибки:
interval в «Функция DateAdd»
Там же в примере:
MsgBox DatAdd(«m», 8, my_date)
Пропущена одна буква — e.
Надеюсь, что помог.

o



admin 2 года ago

<p://vbhack.ru/uroki-vbscript/urok-vbscript-n7-funkcii-dlya-raboty-s-datoy-i-v.html?replyto=21450#respond>

Спасибо за бдительность)



Reply

Preview

Help

Markdown enabled

Name

E-Mail

URL

Код безопасности *



Введите символы отображаемые выше:

Post comment

